EXE-FORMAT SEPARATELY ASSEMBLED MODULES WITH SIMPLIFIED SEGMENT DIRECTIVES





To construct building blocks for large programs, it is often desirable to write programs that call procedures in separately assembled modules. 





When separately assembled exe-format modules with simplified segment directives are linked all code segments are combined into one code segment and all data segments are combined into a single data segment. Thus all procedures in the modules are of type NEAR, and the DS register need to be initialized only once to access all the data. 





Suppose a program calls PROCEDURE1 and PROCEDURE2 in a separately assembled module or in two separately assembled modules. In such a case the EXTERNDEF directive may be used to inform the assembler that the labels PROCEDURE1 and PROCEDURE2 are external. Using one directive, the EXTERNDEF statement, declaring NEAR procedures, has the form:





EXTERNDEF   PROCEDURE1 :  NEAR ,  PROCEDURE2  :  NEAR





Alternatively, separate EXTERNDEF directives may be used:





	EXTERNDEF  PROCEDURE1 : NEAR


	EXTERNDEF  PROCEDURE2 : NEAR





Thus, the program has the following general structure:





		EXTERNDEF  PROCEDURE1 : NEAR ,  PROCEDURE2 : NEAR


	.MODEL SMALL


.DATA


	     .   .   .





	.STACK   0400H 


	.CODE  


	ENTRY_POINT:    


MOV  AX ,   @DATA		; Initialize DS to the segment number of


		MOV  DS ,    AX			; the data segment


		.  .  .


		CALL   PROCEDURE1


		.  .  .


		CALL   PROCEDURE2


		.   .   .


		MOV  AX ,  4C00H			; Return to DOS


		INT  21H				;


	END  ENTRY_POINT





�
If PROCEDUR1 and PROCEDURE2 are in the same separately assembled module, that module is a .asm file with the general structure:





	.MODEL  SMALL


	.CODE 


	     PROCEDURE1  PROC  NEAR


		.  .  .


		RET


	     PROCEDURE1  ENDP





     PROCEDURE2  PROC  NEAR


		.  .  .


		RET


	     PROCEDURE2  ENDP





	END





A separately assembled module  that contains simplified segment directives and which will be called by other modules must end with an END directive which is not followed by an entry-point label. This is because execution starts in the calling module and not in the called module. 





Example: Write an exe-format program which outputs:





THIS IS DISPLAYED BY THE CALLING MODULE


THIS IS DISPLAYED BY THE CALLED MODULE


THIS IS DISPLAYED BY THE CALLING MODULE





The string THIS IS DISPLAYED IN THE CALLING MODULE is defined in the data segment of the calling module. It is to be displayed by a procedure in the calling module. The other string is defined in the data segment of the called module. It is to be displayed by a procedure in that module. 





�
The calling .ASM module is:





	EXTERNDEF  MAIN2 : NEAR


	.MODEL   SMALL


	.STACK    0400H


	.DATA


                 MESSAGE  DB  ‘THIS IS DISPLAYED BY THE CALLING MODULE ’  , 0DH , 0AH , ‘$’


	.CODE  


	     MAIN  PROC


		MOV  AX ,  @DATA			; Initialize DS.


		MOV  DS ,  AX			;


		MOV  DX ,  OFFSET  MESSAGE


		CALL  DISPLAY_STRING		; A call to a NEAR procedure.


		CALL  MAIN2		         ; A call to a NEAR procedure in a separate module


		CALL  DISPLAY_STRING		; A call to a NEAR procedure


		MOV  AX ,  4C00H			; Return to DOS.


		INT  21H				;


	    MAIN  ENDP


    DISPLAY_STRING  PROC


		PUSH  AX


		MOV  AH ,  09H


		INT  21H


		POP  AX


		RET


	   DISPLAY_STRING  ENDP


               END  MAIN





The called .ASM module is:





 	.MODEL   SMALL


       	.DATA


	         STRING  DB  ‘THIS IS DISPLAYED BY THE CALLED MODULE’ , 0DH , 0AH, ‘$’


	.CODE


       		 MAIN2   PROC   NEAR


			PUSH  DX


			MOV  DX ,  OFFSET  STRING


			CALL   STRING_DISPLAY


			POP  DX


			RET


        		 MAIN2    ENDP


         


         		STRING_DISPLAY  PROC  NEAR


			PUSH  AX


			MOV  AH ,  09H


			INT  21H


			POP  AX


			RET


         		STRING_DISPLAY  ENDP


       	END





�
COM-FORMAT SEPARATELY ASSEMBLED MODULES WITH SIMPLIFIED SEGMENT DIRECTIVES





When separately assembled com-format modules with simplified segment directives are linked all code segments are combined into one code segment. Thus all procedures in the modules are of type NEAR.





Example: A module defines a string that is then displayed by a procedure in another module.





The calling module is:





	     EXTERNDEF DISPLAY_STRING : NEAR


.MODEL TINY


.CODE


        	ORG 100H


ENTRY:  JMP L1





      MESSAGE DB 'SOME PEOPLE ARE, UNFOTUNATELY, MORE EQUAL THAN OTHERS', '$'





L1:     	LEA DX , MESSAGE


        	CALL DISPLAY_STRING


        	MOV AX , 4C00H


        	INT 21H


END ENTRY





The called module is:


 


.MODEL  TINY


.CODE


        ORG 100H


   DISPLAY_STRING  PROC  NEAR


        PUSH AX


        MOV AH , 09H


        INT 21H


        POP AX


        RET


   DISPLAY_STRING ENDP


END





�
ASSEMBLING, LINKING, AND EXECUTING SEPARATELY ASSEMBLED MODULES





Suppose PROG1.ASM, PROG2.ASM, PROG3.ASM are three separately assembled modules to be executed as a single program. The steps are:


Assemble each of the files separately, in whatever order. Ignore any warning of a missing stack segment. This will create the object files:


PROG1.OBJ , PROG2.OBJ , and PROG3.OBJ


Link the three object files:


>LINK  PROG1  PROG2   PROG3 , EXEFILE_NAME	     (for exe-format modules)


        or


        >LINK /TINY   PROG1  PROG2  PROG3 , EXEFILE_NAME   (for com-format modules)





.PROG1, PROG2, and PROG3 in the above command lines may appear in any order. If the EXEFILE_NAME is missing, the first object file name in the list is taken as the name of the .EXE file.





Note: If any of the object files is in a different directory from the directory in which the linker is located, then the full path of that file must be given. 


Execute the generated .EXE file by the command:


>EXEFILE_NAME


       A full path may be required.





PASSING VALUES TO OR FROM A PROCEDURE USING THE STACK





Example:     .  .  .


      MOV  DX,  OFFSET  MESSAGE


      PUSH  DX


      CALL  DISPLAY_STRING


      .  .  .


  Here the offset of MESSAGE is passed to the DISPLAY_STRING procedure using the Stack.





This method uses the BP register to access the passed values. By default the BP register points to Stack data, thus it is ideal for accessing procedure parameters. The normal method of accessing parameters in the stack is to copy the value of SP (which always points to the top of the Stack) into BP, and then use offsets from the top of the Stack to access the parameters. In this method, the parameters of a procedure are pushed into the Stack before the call to that procedure.





The memory location at which a parameter is stored is accessed by a memory operand of the form:


	[BP + number]


the segment : offset address of that memory location is:


	SS : (BP + number)





Example: Write an exe-format program, with simplified segment directives, which passes the offset and length of a string, through the Stack, to a procedure DISPLAY_STRING in a separately assembled module. The procedure displays the string using DOS function 02H.





�
Solution: The calling .ASM file is:





EXTERNDEF  DISPLAY_STRING : NEAR


.MODEL SMALL


.STACK 0400H


.DATA


   STRING DB "THIS IS A STRING"


	


.CODE


   START:MOV  AX ,  @DATA


	MOV  DS ,  AX





	MOV AX, OFFSET STRING


	PUSH AX


	MOV AX, LENGTHOF STRING


	PUSH AX


	CALL  DISPLAY_STRING


	MOV  AX ,  4C00H


	INT  21H


END  START





The called .ASM module is:





.MODEL SMALL


.CODE


	


  	DISPLAY_STRING  PROC  NEAR


		PUSH  BP		; Store the current value of BP


		MOV  BP , SP		; Make BP to point to the top of the Stack


		MOV  BX , [BP + 6]	; Copy Offset to BX


		MOV AH, 02H


		MOV CX, [BP + 4]	; Copy length to CX


  	 L2:     MOV DL, [BX]


		INT 21H


		INC BX


		LOOP L2


		POP  BP		; Restore the value of BP


		RET  4			; Discard the two parameters


   	DISPLAY_STRING  ENDP


END





�
Note: The state of the stack after the statement:


	MOV  BP , SP


is:


�
�
                                    high address�
�
�
(Offset)high�
�
�
�
(Offset)low�
[BP + 6]�
�
�
(Length)high�
�
�
�
(Length)low�
[BP + 4]�
�
�
(IP)high�
�
�
�
(IP)low�
�
�
�
(BP)high�
�
�
BP (�
(BP)low�
( SP�
�
�
�
                                       low address�
�



Note: For the RET of the form:


		RET Number


Number = NumberOfParameters * 2 ,  if each parameter is 2 bytes in size. This formula is only valid if the Stack is not used to return values to the calling procedure. This form of RET is used to discard the passed parameters from the Stack on return from the procedure. For example, in the above example if only RET was used instead of  RET 4 ,the state of the Stack would have been:





�
�
                                    high address�
�
�
(Offset)high�
�
�
�
(Offset)low�
�
�
�
(Length)high�
�
�
�
(Length)low�
( SP�
�
�
(IP)high�
�
�
�
(IP)low�
�
�
�
(BP)high�
�
�
�
(BP)low�
                                        low address�
�
�
�
         �
�



To make the portion of the Stack occupied by Offset and Length usable in subsequent PUSH operations, the SP register has to be incremented by 4 in this example. That is achieved by giving the RET instruction the immediate operand 4. The state of the Stack will then be:





�
�
( SP�
�
�
(Offset)high�
                                      high address�
�
�
(Offset)low�
�
�
�
(Length)high�
�
�
�
(Lenght)low�
�
�
�
(IP)high�
�
�
�
(IP)low�
�
�
�
(BP)high�
�
�
�
(BP)low�
                                        low address�
�
�
�
�
�



The Stack may be used to return values from a procedure. 





Example: Write a program fragment, with simplified segment directives, that passes two 16-bit values, through the Stack, to a procedure SUM2 in a separately assembled module. The procedure then returns the sum of the two values through the Stack.





�
Solution: A sample calling .ASM file is:


	


	EXTERNDEF  SUM2 : NEAR


.MODEL SMALL


.STACK  0400H


	.DATA 


	      VAR1  DW  2C3BH


	      VAR2  DW  32B0H


	.CODE  


	START: MOV  AX ,  @DATA


		MOV  DS ,  AX


		PUSH  VAR1


		PUSH  VAR2


		CALL  SUM2


		POP  CX		; Pop the returned value from the Stack.


		.  .  .


		; Manipulate the returned value in CX


		.  .  .


		MOV  AX ,  4C00H


		INT  21H


END  START





The called .ASM module is:





.MODEL SMALL


	.CODE  


	SUM2  PROC  NEAR


		PUSH  BX		;Preserve BX


		PUSH  BP		; Store the current value of BP


		MOV  BP , SP		; Make BP to point to the top of the Stack


		MOV  BX , [BP + 6]	; Copy VAR2 to BX


		ADD  BX , [BP + 8]	; Add VAR1 to BX


		MOV [BP + 8] , BX	; Store the sum in the Stack


		POP  BP			; Restore the value of BP


		POP  BX		;Restore BX


		RET  2			


	SUM2  ENDP


	END





Note: In this example RET 2 is used because the sum stored in the Stack is not to be discarded. That value will be popped in the calling program. 

